# $Id: PKGBUILD 89890 2013-05-04 23:09:02Z alucryd $
# Maintainer: Maxime Gauduin <alucryd at gmail dot com>
# Contributor : sebikul <sebikul@gmail.com>
# Contributor : Massimiliano Torromeo <massimiliano.torromeo@gmail.com>

pkgname=geary
pkgver=0.3.1
pkgrel=4
pkgdesc="A lightweight email client for the GNOME desktop"
arch=('i686' 'x86_64')
url="http://www.yorba.org/projects/geary/"
license=('GPL3')
depends=('desktop-file-utils' 'gmime' 'hicolor-icon-theme' 'libcanberra' 'libgee06' 'libgnome-keyring' 'libnotify' 'libunique3' 'webkitgtk3')
makedepends=('cmake' 'gobject-introspection' 'intltool' 'vala')
install=${pkgname}.install
source=("http://www.yorba.org/download/${pkgname}/${pkgver%.?}/${pkgname}-${pkgver}.tar.xz")
sha256sums=('a1bf8cc95022ce0894a340dea0ab94047ac7a7ea0066b8602b136c74e93ffaed')

build() {
  cd "${srcdir}"/${pkgname}-${pkgver}

  ./configure --prefix=/usr --disable-{desktop-update,icon-update,schemas-compile}
  make
}

package() {
  cd "${srcdir}"/${pkgname}-${pkgver}

  make DESTDIR="${pkgdir}" install
}

# vim: ts=2 sw=2 et:
